

x86 Programming CS 740 Sept. 12, 2007

Topics

- Basics
- Accessing and Moving Data
- Arithmetic operations
- Control Flow
- Procedures
- Data Structures

x86 Processors

Ubiquitous in the desktop, laptop & server markets

Instruction set has evolved over the past ~30 years

- 8086 (1978) was a 16-bit processor
- 386 (1985) extended to 32-bits with a flat address space
 - Capable of running UNIX/Linux
 - 32-bit Linux/gcc uses no instructions introduced in later models
- 64-bit extensions (2003-2004):
 - AMD's x86-64 and Intel's "Intel 64" are nearly identical
 - (not to be confused with Intel's IA-64 in the Itanium machines)

Constraints on the original x86 instruction set:


- Limited memory and silicon space
- Features to facilitate assembly-language programming
 - More recent (RISC) ISAs focus on compiler-generated code

- 2 -


CS 740 F07

Abstract Machines

Machine Model



ASM



Data

- 1) char
- 2) int, float
- 3) double
- 4) struct, array
- 5) pointer


Control

- 1) loops
- 2) conditionals
- 3) goto
- 4) Proc. call
- 5) Proc. return

- 1) byte
- 2) word
- 3) doubleword
- 4) contiguous word allocation
- 5) address of initial byte

CS 740 F07

Assembly Programmer's View



Programmer-Visible State


- **PC** Program Counter
 - Address of next instruction
 - Called "EIP" (IA32) or "RIP" (x86-64)
- **Register File**
 - Heavily used program data
- **Condition Codes**
 - Store status information about most recent arithmetic operation
 - Used for conditional branching
- **Memory**
 - Byte addressable array
 - Code, user data, (some) OS data
 - Includes stack used to support procedures

- 4 -

CS 740 F07

Translation Process

- Code in files p1.c p2.c
- Compile with command: `gcc -o p1.c p2.c -o p`
 - Use optimizations (-O)
 - Put resulting binary in file p



- 5 -

CS 740 F07

Compiling Into Assembly

C Code

```
int sum(int x, int y)
{
    int t = x+y;
    return t;
}
```

Generated IA32 Assembly

```
_sum:
    pushl %ebp
    movl %esp,%ebp
    movl 12(%ebp),%eax
    addl 8(%ebp),%eax
    movl %ebp,%esp
    popl %ebp
    ret
```

Obtain with command

```
gcc -O -S code.c
```

Produces file code.s

- 6 -

CS 740 F07

Object Code

Code for sum

```
0x401040 <sum>:
0x55
0x89
0xe5
0x8b
0x45
0x0c
0x03
0x45
0x08
0x89
0xec
0x5d
0xc3
```

- Total of 13 bytes
- Each instruction 1, 2, or 3 bytes
- Starts at address 0x401040

Assembler

- Translates .s into .o
- Binary encoding of each instruction
- Nearly-complete image of executable code
- Missing linkages between code in different files

Linker

- Resolves references between files
- Combines with static run-time libraries
 - E.g., code for malloc, printf
- Some libraries are *dynamically linked*
 - Linking occurs when program begins execution

- 7 -

CS 740 F07

Machine Instruction Example

```
int t = x+y;
```

```
addl 8(%ebp),%eax
```

Similar to expression:

```
x += y
```

Or

```
int eax;
int *ebp;
eax += ebp[2]
```

```
0x401046: 03 45 08
```

C Code

- Add two signed integers

Assembly

- Add 2 4-byte integers
 - "Long" words in GCC parlance
 - Same instruction whether signed or unsigned

Operands:

- x: Register %eax
- y: Memory M[%ebp+8]
- t: Register %eax
 - » Return function value in %eax

Object Code

- 3-byte instruction
- Stored at address 0x401046

- 8 -

CS 740 F07

Disassembling Object Code

Disassembled

```

00401040 <_sum>:
0:      55                push  %ebp
1:      89 e5             mov   %esp,%ebp
3:      8b 45 0c          mov   0xc(%ebp),%eax
6:      03 45 08          add   0x8(%ebp),%eax
9:      89 ec             mov   %ebp,%esp
b:      5d                pop   %ebp
c:      c3                ret
d:      8d 76 00          lea  0x0(%esi),%esi
    
```

Disassembler

objdump -d p

- Useful tool for examining object code
- Analyzes bit pattern of series of instructions
- Produces approximate rendition of assembly code
- Can be run on either a.out (complete executable) or .o file

- 9 -

CS 740 F07

Alternate Disassembly

Disassembled

Object

```

0x401040:
0x55
0x89
0xe5
0x8b
0x45
0x0c
0x03
0x45
0x08
0x89
0xec
0x5d
0xc3
    
```

```

0x401040 <sum>:      push  %ebp
0x401041 <sum+1>:     mov   %esp,%ebp
0x401043 <sum+3>:     mov   0xc(%ebp),%eax
0x401046 <sum+6>:     add   0x8(%ebp),%eax
0x401049 <sum+9>:     mov   %ebp,%esp
0x40104b <sum+11>:    pop   %ebp
0x40104c <sum+12>:    ret
0x40104d <sum+13>:    lea  0x0(%esi),%esi
    
```

Within gdb Debugger

`gdb p`

`disassemble sum`

- Disassemble procedure

`x/13b sum`

- Examine the 13 bytes starting at sum

- 10 -

CS 740 F07

Moving Data: IA32

Moving Data

`movl Source, Dest;`

- Move 4-byte ("long") word
- Lots of these in typical code

Operand Types

- **Immediate:** Constant integer data
 - Like C constant, but prefixed with '\$'
 - E.g., \$0x400, \$-533
 - Encoded with 1, 2, or 4 bytes
- **Register:** One of 8 integer registers
 - But %esp and %ebp reserved for special use
 - Others have special uses for particular instructions
- **Memory:** 4 consecutive bytes of memory
 - Various "address modes"

%eax
%edx
%ecx
%ebx
%esi
%edi
%esp
%ebp

- 11 -

CS 740 F07

movl Operand Combinations

	Source	Dest	Src, Dest	C Analog
movl	Imm	Reg	movl \$0x4,%eax	temp = 0x4;
		Mem	movl \$-147,(%eax)	*p = -147;
	Reg	Reg	movl %eax,%edx	temp2 = temp1;
		Mem	movl %eax,(%edx)	*p = temp;
Mem	Reg	movl (%eax),%edx	temp = *p;	

Cannot do memory-memory transfer with a single instruction

- 12 -

CS 740 F07

Simple Addressing Modes

Normal (R) Mem[Reg[R]]

- Register R specifies memory address

```
movl (%ecx),%eax
```

Displacement D(R) Mem[Reg[R]+D]

- Register R specifies start of memory region
- Constant displacement D specifies offset

```
movl 8(%ebp),%edx
```

- 13 -

CS 740 F07

Using Simple Addressing Modes

```
void swap(int *xp, int *yp)
{
    int t0 = *xp;
    int t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

swap:

```
pushl %ebp
movl %esp,%ebp
pushl %ebx
```

} Set Up

```
movl 12(%ebp),%ecx
movl 8(%ebp),%edx
movl (%ecx),%eax
movl (%edx),%ebx
movl %eax,(%edx)
movl %ebx,(%ecx)
```

} Body

```
movl -4(%ebp),%ebx
movl %ebp,%esp
popl %ebp
ret
```


} Finish

- 14 -

CS 740 F07

Understanding Swap

```
void swap(int *xp, int *yp)
{
    int t0 = *xp;
    int t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```



Register	Variable
%ecx	yp
%edx	xp
%eax	t1
%ebx	t0

```
movl 12(%ebp),%ecx # ecx = yp
movl 8(%ebp),%edx # edx = xp
movl (%ecx),%eax # eax = *yp (t1)
movl (%edx),%ebx # ebx = *xp (t0)
movl %eax,(%edx) # *xp = eax
movl %ebx,(%ecx) # *yp = ebx
```

- 15 -

CS 740 F07

Indexed Addressing Modes

Most General Form

D(Rb,Ri,S) Mem[Reg[Rb]+S*Reg[Ri]+ D]

- D: Constant "displacement" 1, 2, or 4 bytes
- Rb: Base register: Any of 8 integer registers
- Ri: Index register: Any, except for %esp
 - Unlikely you'd use %ebp, either
- S: Scale: 1, 2, 4, or 8

Special Cases

(Rb,Ri) Mem[Reg[Rb]+Reg[Ri]]

D(Rb,Ri) Mem[Reg[Rb]+Reg[Ri]+D]

(Rb,Ri,S) Mem[Reg[Rb]+S*Reg[Ri]]

- 16 -

CS 740 F07

Address Computation Examples

<code>%edx</code>	<code>0xf000</code>
<code>%ecx</code>	<code>0x100</code>

Expression	Computation	Address
<code>0x8(%edx)</code>	$0xf000 + 0x8$	<code>0xf008</code>
<code>(%edx,%ecx)</code>	$0xf000 + 0x100$	<code>0xf100</code>
<code>(%edx,%ecx,4)</code>	$0xf000 + 4*0x100$	<code>0xf400</code>
<code>0x80(,%edx,2)</code>	$2*0xf000 + 0x80$	<code>0x1e080</code>

- 17 -

CS 740 F07

Address Computation Instruction

`leal Src, Dest`

- *Src* is address mode expression
- Set *Dest* to address denoted by expression

Uses

- Computing addresses without a memory reference
 - E.g., translation of `p = &x[i]`;
- Computing arithmetic expressions of the form $x + k*y$
 - $k = 1, 2, 4, \text{ or } 8$.

- 18 -

CS 740 F07

Some Arithmetic Operations

Format	Computation	
Two Operand Instructions		
<code>addl Src, Dest</code>	$Dest = Dest + Src$	
<code>subl Src, Dest</code>	$Dest = Dest - Src$	
<code>imull Src, Dest</code>	$Dest = Dest * Src$	
<code>sall Src, Dest</code>	$Dest = Dest \ll Src$	Also called <code>shll</code>
<code>sarl Src, Dest</code>	$Dest = Dest \gg Src$	Arithmetic
<code>shrl Src, Dest</code>	$Dest = Dest \gg Src$	Logical
<code>xorl Src, Dest</code>	$Dest = Dest \wedge Src$	
<code>andl Src, Dest</code>	$Dest = Dest \& Src$	
<code>orl Src, Dest</code>	$Dest = Dest Src$	

- 19 -

CS 740 F07

Some Arithmetic Operations

Format	Computation
One Operand Instructions	
<code>incl Dest</code>	$Dest = Dest + 1$
<code>decl Dest</code>	$Dest = Dest - 1$
<code>negl Dest</code>	$Dest = - Dest$
<code>notl Dest</code>	$Dest = \sim Dest$

- 20 -

CS 740 F07

Using leal for Arithmetic Expressions

```

int arith
(int x, int y, int z)
{
    int t1 = x+y;
    int t2 = z+t1;
    int t3 = x+4;
    int t4 = y * 48;
    int t5 = t3 + t4;
    int rval = t2 * t5;
    return rval;
}
    
```

```

arith:
    pushl %ebp
    movl %esp,%ebp
    movl 8(%ebp),%eax
    movl 12(%ebp),%edx
    leal (%edx,%eax),%ecx
    leal (%edx,%edx,2),%edx
    sall $4,%edx
    addl 16(%ebp),%ecx
    leal 4(%edx,%eax),%eax
    imull %ecx,%eax
    movl %ebp,%esp
    popl %ebp
    ret
    
```

Set Up

Body

Finish

- 21 -

CS 740 F07

Understanding arith

```

int arith
(int x, int y, int z)
{
    int t1 = x+y;
    int t2 = z+t1;
    int t3 = x+4;
    int t4 = y * 48;
    int t5 = t3 + t4;
    int rval = t2 * t5;
    return rval;
}
    
```

Stack

•	
•	
•	
16	z
12	y
8	x
4	Rtn adr
0	Old %ebp

Offset

%ebp

```

    movl 8(%ebp),%eax      # eax = x
    movl 12(%ebp),%edx    # edx = y
    leal (%edx,%eax),%ecx # ecx = x+y (t1)
    leal (%edx,%edx,2),%edx # edx = 3*y
    sall $4,%edx         # edx = 48*y (t4)
    addl 16(%ebp),%ecx    # ecx = z+t1 (t2)
    leal 4(%edx,%eax),%eax # eax = 4+t4+x (t5)
    imull %ecx,%eax      # eax = t5*t2 (rval)
    
```

- 22 -

CS 740 F07

Another Example

```

int logical(int x, int y)
{
    int t1 = x^y;
    int t2 = t1 >> 17;
    int mask = (1<<13) - 7;
    int rval = t2 & mask;
    return rval;
}
    
```

```

logical:
    pushl %ebp
    movl %esp,%ebp
    movl 8(%ebp),%eax
    xorl 12(%ebp),%eax
    sarl $17,%eax
    andl $8185,%eax
    movl %ebp,%esp
    popl %ebp
    ret
    
```

Set Up

Body

Finish

$2^{13} = 8192, 2^{13} - 7 = 8185$

```

    movl 8(%ebp),%eax      eax = x
    xorl 12(%ebp),%eax    eax = x^y (t1)
    sarl $17,%eax        eax = t1>>17 (t2)
    andl $8185,%eax      eax = t2 & 8185 (rval)
    
```

- 23 -

CS 740 F07

Data Representations: IA32 + x86-64

Sizes of C Objects (in Bytes)

C Data Type	Typical 32-bit	Intel IA32	x86-64
- unsigned	4	4	4
- int	4	4	4
- long int	4	4	8
- char	1	1	1
- short	2	2	2
- float	4	4	4
- double	8	8	8
- long double	8	10/12	16
- char *	4	4	8

» Or any other pointer

- 24 -

CS 740 F07

x86-64 General Purpose Registers

%rax	%eax	%r8	%r8d
%rdx	%edx	%r9	%r9d
%rcx	%ecx	%r10	%r10d
%rbx	%ebx	%r11	%r11d
%rsi	%esi	%r12	%r12d
%rdi	%edi	%r13	%r13d
%rsp	%esp	%r14	%r14d
%rbp	%ebp	%r15	%r15d

- Extend existing registers. Add 8 new ones.
- Make %ebp/%rbp general purpose

CS 740 F07

Swap in 32-bit Mode (Review)

```
void swap(int *xp, int *yp)
{
    int t0 = *xp;
    int t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

swap:

```
    pushl %ebp
    movl  %esp,%ebp
    pushl %ebx
```

} Set Up

```
    movl 12(%ebp),%ecx
    movl 8(%ebp),%edx
    movl (%ecx),%eax
    movl (%edx),%ebx
    movl %eax,(%edx)
    movl %ebx,(%ecx)
```

} Body

```
    movl -4(%ebp),%ebx
    movl %ebp,%esp
    popl %ebp
    ret
```

} Finish

- 26 -

CS 740 F07

Swap in 64-bit Mode

```
void swap(int *xp, int *yp)
{
    int t0 = *xp;
    int t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

swap:

```
    movl (%rdi), %edx
    movl (%rsi), %eax
    movl %eax, (%rdi)
    movl %edx, (%rsi)
    ret
```

- Operands passed in registers
 - First (xp) in %rdi, second (yp) in %rsi
 - 64-bit pointers
- No stack operations required
- 32-bit data
 - Data held in registers %eax and %edx
 - movl operation

- 27 -

CS 740 F07

Swap Long Ints in 64-bit Mode

```
void swap_l
(long int *xp, long int *yp)
{
    long int t0 = *xp;
    long int t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

swap_l:

```
    movq (%rdi), %rdx
    movq (%rsi), %rax
    movq %rax, (%rdi)
    movq %rdx, (%rsi)
    ret
```

- 64-bit data
 - Data held in registers %rax and %rdx
 - movq operation
 - » "q" stands for quad-word

- 28 -

CS 740 F07

Condition Codes

Single Bit Registers

CF Carry Flag SF Sign Flag
 ZF Zero Flag OF Overflow Flag

Implicitly Set By Arithmetic Operations

addl *Src, Dest* addq *Src, Dest*
 C analog: $t = a + b$ ($a = Src, b = Dest$)

- CF set if carry out from most significant bit
 - Used to detect unsigned overflow
- ZF set if $t == 0$
- SF set if $t < 0$
- OF set if two's complement overflow
 - $(a > 0 \ \&\& \ b > 0 \ \&\& \ t < 0)$
 - $|| \ (a < 0 \ \&\& \ b < 0 \ \&\& \ t >= 0)$

Not set by lea, inc, or dec instructions

- 29 -

CS 740 F07

Setting Condition Codes (cont.)

Explicit Setting by Compare Instruction

cmpl *Src2, Src1* cmpq *Src2, Src1*

- **cmpl b, a** like computing $a - b$ without setting destination
- CF set if carry out from most significant bit
 - Used for unsigned comparisons
- ZF set if $a == b$
- SF set if $(a - b) < 0$
- OF set if two's complement overflow
 - $(a > 0 \ \&\& \ b < 0 \ \&\& \ (a - b) < 0) \ || \ (a < 0 \ \&\& \ b > 0 \ \&\& \ (a - b) > 0)$

- 30 -

CS 740 F07

Setting Condition Codes (cont.)

Explicit Setting by Test instruction

testl *Src2, Src1*
 testq *Src2, Src1*

- Sets condition codes based on value of *Src1* & *Src2*
 - Useful to have one of the operands be a mask
- **testl b, a** like computing $a \& b$ without setting destination
- ZF set when $a \& b == 0$
- SF set when $a \& b < 0$

- 31 -

CS 740 F07

Reading Condition Codes

SetX Instructions

- Set single byte based on combinations of condition codes

SetX	Condition	Description
sete	ZF	Equal / Zero
setne	$\sim ZF$	Not Equal / Not Zero
sets	SF	Negative
setns	$\sim SF$	Nonnegative
setg	$\sim (SF \wedge OF) \ \& \ \sim ZF$	Greater (Signed)
setge	$\sim (SF \wedge OF)$	Greater or Equal (Signed)
setl	$(SF \wedge OF)$	Less (Signed)
setle	$(SF \wedge OF) \ \ ZF$	Less or Equal (Signed)
seta	$\sim CF \ \& \ \sim ZF$	Above (unsigned)
setb	CF	Below (unsigned)

- 32 -

CS 740 F07

Reading Condition Codes (Cont.)

SetX Instructions

- Set single byte based on combinations of condition codes
- One of 8 addressable byte registers
 - Embedded within first 4 integer registers
 - Does not alter remaining 3 bytes
 - Typically use movzbl to finish job

```
int gt (int x, int y)
{
    return x > y;
}
```

Body

```
movl 12(%ebp),%eax # eax = y
cmpl %eax,8(%ebp) # Compare x : y
setg %al          # al = x > y
movzbl %al,%eax  # Zero rest of %eax
```

Note
inverted
ordering!

%eax	%ah	%al
%edx	%dh	%dl
%ecx	%ch	%cl
%ebx	%bh	%bl
%esi		
%edi		
%esp		
%ebp		

- 33 -

CS 740 F07

Reading condition codes: x86-64

SetX Instructions

- Set single byte based on combinations of condition codes
 - Does not alter remaining 7 bytes

```
int gt (long x, long y)
{
    return x > y;
}
```

```
long lgt (long x, long y)
{
    return x > y;
}
```

- x86-64 arguments
 - x in %rdi
 - y in %rsi

Body (same for both)

(32-bit instructions set high order 32 bits to 0)

```
xorl %eax, %eax # eax = 0
cmpq %rsi, %rdi # Compare x : y
setg %al        # al = x > y
```

- 34 -

CS 740 F07

Jumping

jX Instructions

- Jump to different part of code depending on condition codes

jX	Condition	Description
jmp	1	Unconditional
je	ZF	Equal / Zero
jne	~ZF	Not Equal / Not Zero
js	SF	Negative
jns	~SF	Nonnegative
jg	~(SF^OF) & ~ZF	Greater (Signed)
jge	~(SF^OF)	Greater or Equal (Signed)
jl	(SF^OF)	Less (Signed)
jle	(SF^OF) ZF	Less or Equal (Signed)
ja	~CF & ~ZF	Above (unsigned)
jb	CF	Below (unsigned)

- 35 -

CS 740 F07

Conditional Branch Example

```
int absdiff(
    int x, int y)
{
    int result;
    if (x > y) {
        result = x-y;
    } else {
        result = y-x;
    }
    return result;
}
```

```
absdiff:
    pushl %ebp
    movl %esp, %ebp
    movl 8(%ebp), %edx
    movl 12(%ebp), %eax
    cmpl %eax, %edx
    jle .L7
    subl %eax, %edx
    movl %edx, %eax
.L8:
    leave
    ret
.L7:
    subl %edx, %eax
    jmp .L8
```

} Set Up
} Body1
} Finish
} Body2

- 36 -

CS 740 F07

Conditionals: x86-64

```
int absdiff(          absdiff: # x in %edi, y in %esi
  int x, int y)
{
  int result;
  if (x > y) {
    result = x-y;
  } else {
    result = y-x;
  }
  return result;
}
```

```
movl %edi, %eax # v = x
movl %esi, %edx # ve = y
subl %esi, %eax # v -= y
subl %edi, %edx # ve -= x
cmpl %esi, %edi # x:y
cmovle %edx, %eax # v=ve if <=
ret
```

- Conditional move instruction
 - `cmovC src, dest`
 - Move value from `src` to `dest` if condition `C` holds
 - More efficient than conditional branching
 - » Simple & predictable control flow

- 37 -

CS 740 F07

General Form with Conditional Move

C Code

```
val = Test ? Then-Expr : Else-Expr;
```

Conditional Move Version

```
val = Then-Expr;
vale = Else-Expr;
val = vale if !Test;
```

- Both values get computed
- Overwrite then-value with else-value if condition doesn't hold

- 38 -

CS 740 F07

Limitations of Conditional Move

```
val = Then-Expr;
vale = Else-Expr;
val = vale if !Test;
```

```
int xgty = 0, xltey = 0;

int absdiff_se(
  int x, int y)
{
  int result;
  if (x > y) {
    xgty++; result = x-y;
  } else {
    xltey++; result = y-x;
  }
  return result;
}
```

Don't use when:

- Then-Expr or Else-Expr has side effect
- Then-Expr or Else-Expr requires significant computation

- 39 -

CS 740 F07

"Do-While" Loop Example

C Code

```
int fact_do(int x)
{
  int result = 1;
  do {
    result *= x;
    x = x-1;
  } while (x > 1);

  return result;
}
```

Goto Version

```
int fact_goto(int x)
{
  int result = 1;
loop:
  result *= x;
  x = x-1;
  if (x > 1)
    goto loop;
  return result;
}
```

- Use backward branch to continue looping
- Only take branch when "while" condition holds

- 40 -

CS 740 F07

"Do-While" Loop Compilation

Goto Version

```
int
fact_goto(int x)
{
    int result = 1;

loop:
    result *= x;
    x = x-1;
    if (x > 1)
        goto loop;

    return result;
}
```

Assembly

```
fact_goto:
    pushl %ebp          # Setup
    movl %esp,%ebp     # Setup
    movl $1,%eax       # eax = 1
    movl 8(%ebp),%edx  # edx = x

L11:
    imull %edx,%eax    # result *= x
    decl %edx          # x--
    cmpl $1,%edx       # Compare x : 1
    jg L11              # if > goto loop

    movl %ebp,%esp     # Finish
    popl %ebp          # Finish
    ret                # Finish
```

Registers

```
%edx  x
%eax  result
```

- 41 -

CS 740 F07

"While" Loop Translation

C Code

```
while (Test)
    Body
```

Do-While Version

```
if (!Test)
    goto done;
do
    Body
while (Test);
done:
```

Goto Version

```
if (!Test)
    goto done;
loop:
    Body
    if (Test)
        goto loop;
done:
```

- 42 -

CS 740 F07

"For" → "While" → "Do-While"

For Version

```
for (Init; Test; Update)
    Body
```

While Version

```
Init;
while (Test) {
    Body
    Update ;
}
```

Do-While Version

```
Init;
if (!Test)
    goto done;
do {
    Body
    Update ;
} while (Test)
done:
```

Goto Version

```
Init;
if (!Test)
    goto done;
loop:
    Body
    Update ;
    if (Test)
        goto loop;
done:
```

- 43 -

CS 740 F07

Switch Statements

Implementation Options

- Series of conditionals
 - Organize in tree structure
 - Logarithmic performance
- Jump Table
 - Lookup branch target
 - Constant time
 - Possible when cases are small integer constants
- GCC
 - Picks one based on case structure

- 44 -

CS 740 F07

Jump Table Structure


Switch Form

```
switch(x) {
  case val_0:
    Block 0
  case val_1:
    Block 1
    . . .
  case val_n-1:
    Block n-1
}
```

Jump Table

jtab:	Targ0
	Targ1
	Targ2
	.
	.
	Targn-1

Jump Targets



Approx. Translation

```
target = JTab[x];
goto *target;
```

- 45 -

CS 740 F07

Switch Statement Example

```
long switch_eg
(long x, long y, long z)
{
  long w = 1;
  switch(x) {
    . . .
  }
  return w;
}
```

Setup:

```
switch_eg:
  pushl %ebp          # Setup
  movl %esp, %ebp    # Setup
  pushl %ebx          # Setup
  movl $1, %ebx       # w = 1
  movl 8(%ebp), %edx  # edx = x
  movl 16(%ebp), %ecx # ecx = z
  cmpl $6, %edx       # x:6
  ja .L61             # if > goto default
  jmp *.L62(, %edx, 4) # goto JTab[x]
```

- 46 -

CS 740 F07

Procedure Calls

x86 (IA32):

- stack discipline

X86-64:


- argument passing in registers

- 47 -

CS 740 F07

IA32 Stack

- Region of memory managed with stack discipline
- Grows toward lower addresses
- Register `%esp` indicates lowest stack address
 - address of top element




- 48 -

CS 740 F07

IA32 Stack Pushing

Pushing

- `pushl Src`
- Fetch operand at `Src`
- Decrement `%esp` by 4
- Write operand at address given by `%esp`




- 49 -

CS 740 F07

IA32 Stack Popping

Popping

- `popl Dest`
- Read operand at address given by `%esp`
- Increment `%esp` by 4
- Write to `Dest`



- 50 -

CS 740 F07

Procedure Control Flow

- Use stack to support procedure call and return

Procedure call:

`call label` Push return address on stack; Jump to `label`

Return address value

- Address of instruction beyond `call`

- Example from disassembly

```
804854e: e8 3d 06 00 00 call 8048b90 <main>
8048553: 50          pushl %eax
```

- Return address = `0x8048553`

Procedure return:


- `ret` Pop address from stack; Jump to address

- 51 -

CS 740 F07

Procedure Call Example


```
804854e: e8 3d 06 00 00 call 8048b90 <main>
8048553: 50          pushl %eax
                        call 8048b90
```



- 52 -

CS 740 F07

Procedure Return Example



Stack-Based Languages


Languages that support recursion

- e.g., C, Pascal


Stack Allocated in Frames

- state for procedure invocation
- return point, arguments, locals

Code Example



Stack (grows down)




IA32/Linux Stack Frame

Current Stack Frame ("Top" to Bottom)

- Parameters for function about to call
 - "Argument build"
- Local variables
 - If can't keep in registers
- Saved register context
- Old frame pointer

Caller Stack Frame

- Return address
 - Pushed by call instruction
- Arguments for this call



Revisiting swap


```
int zip1 = 15213;
int zip2 = 91125;
```

```
void call_swap()
{
    swap(&zip1, &zip2);
}
```

```
void swap(int *xp, int *yp)
{
    int t0 = *xp;
    int t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

Calling swap from call_swap

```
call_swap:
    . . .
    pushl $zip2 # Global Var
    pushl $zip1 # Global Var
    call swap
    . . .
```



Revisiting swap

```
void swap(int *xp, int *yp)
{
    int t0 = *xp;
    int t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

```
swap:
    pushl %ebp
    movl %esp,%ebp
    pushl %ebx
    movl 12(%ebp),%ecx
    movl 8(%ebp),%edx
    movl (%ecx),%eax
    movl (%edx),%ebx
    movl %eax,(%edx)
    movl %ebx,(%ecx)
    movl -4(%ebp),%ebx
    movl %ebp,%esp
    popl %ebp
    ret
```

Set Up

Body


Finish

- 57 -


CS 740 F07

swap Setup #1

Entering Stack



Resulting Stack




```
swap:
    pushl %ebp
    movl %esp,%ebp
    pushl %ebx
```

- 58 -


CS 740 F07

swap Setup #2

Entering Stack



Resulting Stack




```
swap:
    pushl %ebp
    movl %esp,%ebp
    pushl %ebx
```

- 59 -


CS 740 F07

swap Setup #3

Entering Stack



Resulting Stack




```
swap:
    pushl %ebp
    movl %esp,%ebp
    pushl %ebx
```


- 60 -

CS 740 F07


Effect of swap Setup




swap Finish #1




swap Finish #2



swap Finish #3



swap Finish #4



Observation

- Saved & restored register `%ebx`
- Didn't do so for `%eax`, `%ecx`, or `%edx`

```
movl -4(%ebp),%ebx
movl %ebp,%esp
popl %ebp
ret
```

- 65 -

CS 740 F07

Register Saving Conventions

When procedure `yoo` calls `who`:

- `yoo` is the *caller*, who is the *callee*

Can Register be Used for Temporary Storage?

```
yoo:
. . .
movl $15213, %edx
call who
addl %edx, %eax
. . .
ret
```

```
who:
. . .
movl 8(%ebp), %edx
addl $91125, %edx
. . .
ret
```

- Contents of register `%edx` overwritten by `who`

- 66 -

CS 740 F07

Register Saving Conventions

When procedure `yoo` calls `who`:

- `yoo` is the *caller*, who is the *callee*

Can Register be Used for Temporary Storage?

Conventions

- "Caller Save"
 - Caller saves temporary in its frame before calling
- "Callee Save"
 - Callee saves temporary in its frame before using


- 67 -

CS 740 F07

IA32/Linux Register Usage

Integer Registers

- Two have special uses
`%ebp`, `%esp`
- Three managed as callee-save
`%ebx`, `%esi`, `%edi`
- Old values saved on stack prior to using
- Three managed as caller-save
`%eax`, `%edx`, `%ecx`
- Do what you please, but expect any callee to do so, as well
- Register `%eax` also stores returned value



- 68 -

CS 740 F07

Recursive Factorial

```
int rfact(int x)
{
  int rval;
  if (x <= 1)
    return 1;
  rval = rfact(x-1);
  return rval * x;
}
```

```
.globl rfact
.type
rfact,@function
rfact:
  pushl %ebp
  movl %esp,%ebp
  pushl %ebx
  movl 8(%ebp),%ebx
  cmpl $1,%ebx
  jle .L78
  leal -1(%ebx),%eax
  pushl %eax
  call rfact
  imull %ebx,%eax
  jmp .L79
  .align 4
.L78:
  movl $1,%eax
.L79:
  movl -4(%ebp),%ebx
  movl %ebp,%esp
  popl %ebp
  ret
```

Registers

- `%eax` used without first saving
- `%ebx` used, but save at beginning & restore at end

- 69 - CS 740 F07

x86-64 General Purpose Registers

<code>%rax</code>	<code>%eax</code>	<code>%r8</code>	<code>%r8d</code>
<code>%rbx</code>	<code>%ebx</code>	<code>%r9</code>	<code>%r9d</code>
<code>%rcx</code>	<code>%ecx</code>	<code>%r10</code>	<code>%r10d</code>
<code>%rdx</code>	<code>%edx</code>	<code>%r11</code>	<code>%r11d</code>
<code>%rsi</code>	<code>%esi</code>	<code>%r12</code>	<code>%r12d</code>
<code>%rdi</code>	<code>%edi</code>	<code>%r13</code>	<code>%r13d</code>
<code>%rsp</code>	<code>%esp</code>	<code>%r14</code>	<code>%r14d</code>
<code>%rbp</code>	<code>%ebp</code>	<code>%r15</code>	<code>%r15d</code>

- Twice the number of registers
- Accessible as 8, 16, 32, or 64 bits

- 70 - CS 740 F07

x86-64 Register Conventions

<code>%rax</code>	Return Value	<code>%r8</code>	Argument #5
<code>%rbx</code>	Callee Saved	<code>%r9</code>	Argument #6
<code>%rcx</code>	Argument #4	<code>%r10</code>	Callee Saved
<code>%rdx</code>	Argument #3	<code>%r11</code>	Used for linking
<code>%rsi</code>	Argument #2	<code>%r12</code>	C: Callee Saved
<code>%rdi</code>	Argument #1	<code>%r13</code>	Callee Saved
<code>%rsp</code>	Stack Pointer	<code>%r14</code>	Callee Saved
<code>%rbp</code>	Callee Saved	<code>%r15</code>	Callee Saved

- 71 - CS 740 F07

x86-64 Registers

Arguments passed to functions via registers

- If more than 6 integral parameters, then pass rest on stack
- These registers can be used as caller-saved as well

All References to Stack Frame via Stack Pointer

- Eliminates need to update `%ebp`

Other Registers

- 6+1 callee saved
- 2 or 3 have special uses

- 72 - CS 740 F07

Basic Data Types

Integral

- Stored & operated on in general registers
- Signed vs. unsigned depends on instructions used

Intel	GAS	Bytes	C
byte	b	1	[unsigned] char
word	w	2	[unsigned] short
double word	l	4	[unsigned] int
quad word	q	8	[unsigned] long int (x86-64)

Floating Point

- Stored & operated on in floating point registers

Intel	GAS	Bytes	C
Single	s	4	float
Double	l	8	double
Extended	t	10/12/16	long double

- 73 -

CS 740 F07

Array Accessing Example

Computation

- Register %edx contains starting address of array
- Register %eax contains array index
- Desired digit at $4 * \text{eax} + \text{edx}$
- Use memory reference (`%edx,%eax,4`)

```
int get_digit
(zip_digit z, int dig)
{
    return z[dig];
}
```

IA32 Memory Reference Code

```
# %edx = z
# %eax = dig
movl (%edx,%eax,4),%eax # z[dig]
```

- 74 -


CS 740 F07

Structures

```
struct rec {
    int i;
    int a[3];
    int *p;
};
```

```
void
set_i(struct rec *r,
      int val)
{
    r->i = val;
}
```

Memory Layout



IA32 Assembly


```
# %eax = val
# %edx = r
movl %eax,(%edx) # Mem[r] = val
```

- 75 -

CS 740 F07

Generating Pointer to Struct. Member

```
struct rec {
    int i;
    int a[3];
    int *p;
};
```



Generating Pointer to Array Element

- Offset of each structure member determined at compile time

```
int *
find_a
(struct rec *r, int idx)
{
    return &r->a[idx];
}
```

```
# %ecx = idx
# %edx = r
leal 0,(%ecx,4),%eax # 4*idx
leal 4(%eax,%edx),%eax # r+4*idx+4
```

- 76 -


CS 740 F07

Structure Referencing (Cont.)

C Code

```
struct rec {  
    int i;  
    int a[3];  
    int *p;  
};
```

```
void  
set_p(struct rec *r)  
{  
    r->p =  
    &r->a[r->i];  
}
```



```
# %edx = r  
movl (%edx),%ecx # r->i  
leal 0(,%ecx,4),%eax # 4*(r->i)  
leal 4(%edx,%eax),%eax # r+4+4*(r->i)  
movl %eax,16(%edx) # Update r->p
```